# $Id: PKGBUILD 66382 2012-02-24 12:03:33Z lfleischer $
# Maintainer: Lukas Fleischer <archlinux at cryptocrack dot de>
# Contributor: Eric Belanger <eric@archlinux.org>
# Contributor: Darwin Bautista <djclue917@gmail.com>
# Contributor: Bob Finch <w9ya@qrparci.net>

pkgname=portaudio
pkgver=19_20111121
pkgrel=1
pkgdesc='A free, cross-platform, open source, audio I/O library.'
arch=('i686' 'x86_64')
url='http://www.portaudio.com/'
license=('custom')
depends=('jack')
options=('!libtool')
source=("http://www.portaudio.com/archives/pa_stable_v$pkgver.tgz")
md5sums=('25c85c1cc5e9e657486cbc299c6c035a')

build() {
  cd "$srcdir/$pkgname"

  ./configure --prefix=/usr
  make
}

package() {
  cd "$srcdir/$pkgname"

  make DESTDIR="$pkgdir" install
  install -Dm0644 LICENSE.txt "$pkgdir/usr/share/licenses/$pkgname/LICENSE.txt"
}
